%% US Public Debt and Safe Asset Market Power
%% Jason Choi, Rishabh Kirpalani, and Diego Perez
%% Nov 24, 2024

%% Solve Competitive Equilibrium

%----------------------------------------------------------------
% 0. Housekeeping
%----------------------------------------------------------------

close all

%----------------------------------------------------------------
% 1. Defining variables
%----------------------------------------------------------------

// Endogenous Variables
var b rb rkstar rk krw_star kus_star krw kus kstar k wstar w c_rw c_us drdb spread vrw vus dMrwdb by;

// Exogenous Variables
var nnu oomega A Astar eeta;

// Shocks
varexo eps_nnu eps_oomega eps_A eps_eeta;

// Parameters
parameters ggamma bbeta llambda aalpha Astarbar Abar iiota iiota_star ddelta_rw ddelta_us
  nnu_bar oomega_bar rrho_nnu rrho_oomega ssigma_nnu ssigma_oomega rrho_A ssigma_A rrho_Astar ssigma_Astar
  b_ce rb_ce rkstar_ce rk_ce krw_star_ce kus_star_ce krw_ce kus_ce kstar_ce k_ce
  capKstar_ce capK_ce wstar_ce w_ce crw_ce cus_ce spread_ce vrw_ce vus_ce eeta_bar ssigma_eeta beta_alpha beta_beta beta_mean beta_var beta_sd;

%----------------------------------------------------------------
% 2. Calibration
%----------------------------------------------------------------

% // Parameters
disp('% Epsilon = 2.2, Lambda = 1') 

ggamma = 2;
bbeta = 0.9886;
% eeta = 0.545;
llambda = 1;
aalpha = 0.3;
Astarbar = 0.9254;
Abar = 0.8154;
iiota = 0.9070;
iiota_star = 0.7939;
ddelta_rw = 0.1;
ddelta_us = 0.1;
nnu_bar = 0.0048;
oomega_bar = 0.0025;
rrho_nnu = 0.99;
ssigma_nnu = 0.01;
rrho_oomega = 0.95;
ssigma_oomega = 0.48;
rrho_A = 0.95;
ssigma_A = 0.02;
rrho_Astar = rrho_A;
ssigma_Astar = ssigma_A;

eeta_bar = 0.545;

beta_alpha = 0.001;
beta_beta = beta_alpha/eeta_bar - beta_alpha;
beta_mean = beta_alpha/(beta_alpha+beta_beta);
beta_var = (beta_alpha*beta_beta)/((beta_alpha+beta_beta)^2*(beta_alpha+beta_beta+1));
beta_sd = sqrt(beta_var);
ssigma_eeta = beta_sd;

% Analytic Steady State (Competitive Equilibrium)
b_ce = (oomega_bar/nnu_bar)^(1/(eeta_bar-1-llambda));
rb_ce = 1/bbeta - nnu_bar*(b_ce)^(eeta_bar-1) - 1;
rkstar_ce = 1/bbeta	+ ddelta_rw - 1;
rk_ce = 1/bbeta	+ ddelta_us - 1;
krw_star_ce = ((aalpha*(1-iiota_star)*Astarbar*((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^((aalpha*(1-iiota_star)-1)/(aalpha*(1-iiota_star))))/(1/bbeta+ddelta_us-1))^((aalpha*(1-iiota_star))/(1-aalpha));
kus_star_ce = ((1/bbeta+ddelta_rw-1)/(aalpha*iiota_star*Astarbar))^(1/(aalpha*(1-iiota_star)))*krw_star_ce^((1-iiota_star*aalpha)/(aalpha*(1-iiota_star)));
krw_ce = ((aalpha*iiota*Abar*((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^((aalpha*iiota-1)/(aalpha*iiota)))/(1/bbeta+ddelta_rw-1))^((aalpha*iiota)/(1-aalpha));
kus_ce = ((1/bbeta+ddelta_us-1)/(aalpha*(1-iiota)*Abar))^(1/(aalpha*iiota))*krw_ce^((1-(1-iiota)*aalpha)/(aalpha*iiota));
kstar_ce = krw_star_ce + krw_ce;
k_ce = kus_star_ce + kus_ce;
capKstar_ce = krw_star_ce^iiota_star*kus_star_ce^(1-iiota_star);
capK_ce = krw_ce^(1-iiota)*kus_ce^iiota;
wstar_ce = Astarbar*(1-aalpha)*(capKstar_ce)^aalpha;
w_ce = Abar*(1-aalpha)*(capK_ce)^aalpha;
crw_ce = wstar_ce + (rkstar_ce-ddelta_rw)*kstar_ce + nnu_bar/eeta_bar*(b_ce)^eeta_bar + rb_ce*b_ce;
cus_ce = w_ce + (rk_ce-ddelta_us)*k_ce - oomega_bar/(1+llambda)*(b_ce)^(1+llambda) - rb_ce*b_ce;
drdb_ce = 0;
dMrwdb_ce = 0;
nnu_ce = nnu_bar;
oomega_ce = oomega_bar;
A_ce = Abar;
Astar_ce = Astarbar;
spread_ce = (rk_ce-ddelta_us-rb_ce);
vrw_ce = crw_ce^(1-ggamma)/(1-ggamma)/(1-bbeta);
vus_ce = cus_ce^(1-ggamma)/(1-ggamma)/(1-bbeta);
by_ce = b_ce/(A_ce*(kus_ce^iiota*krw_ce^(1-iiota))^aalpha);
eeta_me = eeta_bar;

%----------------------------------------------------------------
% 3. Model
%----------------------------------------------------------------

model;

c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(nnu*b^(eeta-1)+1+rb);
c_rw^(-ggamma) = bbeta*c_rw(+1)^(-ggamma)*(1-ddelta_rw+rkstar);
c_rw + kstar + b = wstar + (1-ddelta_rw+rkstar(-1))*kstar(-1) + nnu(-1)/eeta*(b(-1))^eeta + (1+rb(-1))*b(-1);

c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(oomega*b^llambda+1+rb+drdb*b);
c_us^(-ggamma) = bbeta*c_us(+1)^(-ggamma)*(1-ddelta_us+rk);
c_us + k - b = w + (1-ddelta_us+rk(-1))*k(-1) - oomega(-1)/(1+llambda)*(b(-1))^(1+llambda) - (1+rb(-1))*b(-1);

rk = Astar*aalpha*(1-iiota_star)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star)-1);
rkstar = Astar*aalpha*iiota_star*krw_star^(aalpha*iiota_star-1)*kus_star^(aalpha*(1-iiota_star));
rk = A*aalpha*iiota*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota-1);
rkstar = A*aalpha*(1-iiota)*krw^(aalpha*(1-iiota)-1)*kus^(aalpha*iiota);
wstar = Astar*(1-aalpha)*krw_star^(aalpha*iiota_star)*kus_star^(aalpha*(1-iiota_star));
w = A*(1-aalpha)*krw^(aalpha*(1-iiota))*kus^(aalpha*iiota);

drdb = 0;
dMrwdb = 0;

k = kus + kus_star;
kstar = krw + krw_star;

eeta = eeta_bar + ssigma_eeta*eps_eeta;

log(nnu) = (1-rrho_nnu)*log(nnu_bar) + rrho_nnu*log(nnu(-1)) + ssigma_nnu*eps_nnu;
log(oomega) = (1-rrho_oomega)*log(oomega_bar) + rrho_oomega*log(oomega(-1)) + ssigma_oomega*eps_oomega;
log(A) = (1-rrho_A)*log(Abar) + rrho_A*log(A(-1)) + ssigma_A*eps_A;
log(Astar) = (1-rrho_Astar)*log(Astarbar) + rrho_Astar*log(Astar(-1)) + ssigma_Astar*eps_A;

spread = (rk-ddelta_us-rb);

vrw = c_rw^(1-ggamma)/(1-ggamma) + bbeta*vrw(+1);
vus = c_us^(1-ggamma)/(1-ggamma) + bbeta*vus(+1);

by = b/(A*(kus^iiota*krw^(1-iiota))^aalpha);

end;

%----------------------------------------------------------------
% 4. Computation
%----------------------------------------------------------------

initval;
  b = b_ce;
  rb = rb_ce;
  rkstar = rkstar_ce;
  rk = rk_ce;
  krw_star = krw_star_ce;
  kus_star = kus_star_ce;
  krw = krw_ce;
  kus = kus_ce;
  kstar = kstar_ce;
  k = k_ce;
  wstar = wstar_ce;
  w = w_ce;
  c_rw = crw_ce;
  c_us = cus_ce;
  drdb = drdb_ce;
  dMrwdb = dMrwdb_ce;
  nnu = nnu_ce;
  oomega = oomega_ce;
  spread = spread_ce;
  A = A_ce;
  Astar = Astar_ce;
  vrw = vrw_ce;
  vus = vus_ce;
  by = by_ce;
  eeta = eeta_me;
end;

resid;
check;

shocks;
  var eps_nnu = 1;
  var eps_oomega = 1;
  var eps_A = 1;
  var eps_eeta = 1;
end;

set_dynare_seed('default');
stoch_simul(order=2,nograph,noprint,periods=10000,drop=1000,pruning);

%----------------------------------------------------------------
% 5. Generate moments
%----------------------------------------------------------------

spread_path = (rk-ddelta_us-rb)*100;
var_b = var(b);
var_sp = var(spread_path);
auto_b = autocorr(b);
auto_sp = autocorr(spread_path);
cost = oomega./(1+llambda).*(b).^(1+llambda);
benefit = nnu./eeta.*(b).^eeta;
cost_p = oomega.*(b).^(llambda);
benefit_p = nnu.*(b).^(eeta-1);
corr_pq = corr(spread_path,b);

moments = [mean(b) mean(spread_path) var_b var_sp corr_pq auto_b(2) auto_sp(2)]';
data_mom = [0.41 0.62 0.03 0.086 -0.56 0.96 0.70]';
rowNames = {'Mean b','Mean sp','Var b','Var sp','Corr (b,sp)','Autocorr b','Autocorr sp'};
colNames = {'Model Moments','Data Moments'};
TableA0 = array2table([moments data_mom],'RowNames',rowNames,'VariableNames',colNames)

deficit =  cost(1:end-1) + b(2:end) - (1+rb(1:end-1)).*b(1:end-1);
ca = -(b(2:end)-b(1:end-1)) + kus_star(2:end)-kus_star(1:end-1) - (krw(2:end)-krw(1:end-1));
nfa = - b + kus_star - krw;
var_ca = var(ca);
var_nfa = var(nfa);
var_deficit = var(deficit);
corr_nfa_b = corr(nfa,b);
corr_ca_def = corr(ca,deficit);
var_cost = var(cost);
var_benefit = var(benefit);
var_cost_p = var(cost_p);
var_benefit_p = var(benefit_p);

%----------------------------------------------------------------
% 6. Calculate welfare from transition
%---------------------------------------------------------------
b_pos = strmatch('b',M_.endo_names,'exact');
rb_pos = strmatch('rb',M_.endo_names,'exact');
rkstar_pos = strmatch('rkstar',M_.endo_names,'exact');
rk_pos = strmatch('rk',M_.endo_names,'exact');
krw_star_pos = strmatch('krw_star',M_.endo_names,'exact');
kus_star_pos = strmatch('kus_star',M_.endo_names,'exact');
krw_pos = strmatch('krw',M_.endo_names,'exact');
kus_pos = strmatch('kus',M_.endo_names,'exact');
kstar_pos = strmatch('kstar',M_.endo_names,'exact');
k_pos = strmatch('k',M_.endo_names,'exact');
wstar_pos = strmatch('wstar',M_.endo_names,'exact');
w_pos = strmatch('w',M_.endo_names,'exact');
crw_pos = strmatch('c_rw',M_.endo_names,'exact');
cus_pos = strmatch('c_us',M_.endo_names,'exact');
drdb_pos = strmatch('drdb',M_.endo_names,'exact');
dMrwdb_pos = strmatch('dMrwdb',M_.endo_names,'exact');
vrw_pos = strmatch('vrw',M_.endo_names,'exact');
vus_pos = strmatch('vus',M_.endo_names,'exact');
nnu_pos = strmatch('nnu',M_.endo_names,'exact');
oomega_pos = strmatch('oomega',M_.endo_names,'exact');
A_pos = strmatch('A',M_.endo_names,'exact');
Astar_pos = strmatch('Astar',M_.endo_names,'exact');
z_pos = strmatch('z',M_.endo_names,'exact');
by_pos = strmatch('by',M_.endo_names,'exact');
eeta_pos = strmatch('eeta',M_.endo_names,'exact');

shock_matrix = randn(size(oo_.endo_simul,2),M_.exo_nbr); %create shock matrix with number of time periods in rows

shock_matrix(:,4) = (betarnd(beta_alpha,beta_beta,size(oo_.endo_simul,2),1) - eeta_bar)/beta_sd; %create shock matrix with number of time periods in rows
histogram(shock_matrix(:,4).*beta_sd+eeta_bar)

sim_y_forward = simult_(M_,options_,oo_.mean,oo_.dr,shock_matrix,options_.order);

z = sim_y_forward(oomega_pos,:); 
b_path = sim_y_forward(by_pos,:);
FVD = sim_y_forward(eeta_pos,:)<eeta_bar;
eeta_path = sim_y_forward(eeta_pos,:);
spread_path = (sim_y_forward(rk_pos,:)-ddelta_us-sim_y_forward(rb_pos,:))*100;

spread_path(b_path<0)=[];
z(b_path<0)=[];
FVD(b_path<0)=[];
eeta_path(b_path<0)=[];
b_path(b_path<0)=[];

b_path(spread_path<0)=[];
z(spread_path<0)=[];
FVD(spread_path<0)=[];
eeta_path(spread_path<0)=[];
spread_path(spread_path<0)=[];

XX = log(b_path);
YY = spread_path;

XXp = [log(b_path)' FVD' log(b_path)'.*FVD'];

table = fitlm(XXp,YY)
low_beta = table2array(table.Coefficients(2,1));
high_beta = table2array(table.Coefficients(2,1))+table2array(table.Coefficients(4,1));
low_elast = 1/(low_beta/mean(spread_path));
high_elast = 1/(high_beta/mean(spread_path));

model_mom = [high_elast' low_elast']';
data_mom = [-1.45 -3.23]';
rowNames = {'High Vol Elasticity','Low Vol Elasticity'};
colNames = {'Model Moments','Data Moments'};
TableA0 = array2table([model_mom data_mom],'RowNames',rowNames,'VariableNames',colNames)

writematrix([XXp YY' z' eeta_path'],'simulated_ce.xlsx')
histogram(eeta_path)

%----------------------------------------------------------------
% 6. Calculate welfare from transition
%---------------------------------------------------------------

b_ce_eta = mean(b);
rb_ce_eta = mean(rb);
spread_ce_eta = mean(spread);

oo_ce_eta = oo_;
M_ce_eta = M_;
options_ce_eta = options_;  

save ce_eta_save oo_ce_eta vus_ce vrw_ce cus_ce crw_ce M_ce_eta options_ce_eta b_ce_eta rb_ce_eta spread_ce_eta;

